
1. Objectives 

 

To get familiar with Bluetooth wireless technology. To modify the data acquisition system 

developed in the previous lab to use an HC-06 Bluetooth module to transmit the temperature data 

wirelessly to the host PC.  

 

2. Introduction  

 

2.1. Background on Bluetooth Wireless Technology 

 

Bluetooth is a wireless technology standard for exchanging data over short distances (using 

short-wavelength UHF radio waves in the ISM band from 2.4 to 2.485 GHz) from fixed and 

mobile devices, and building personal area networks (PANs). Invented by telecom vendor 

Ericsson in 1994, it was originally conceived as a wireless alternative to RS-232 data cables. It 

can connect several devices, overcoming problems of synchronization [1]. In short, Bluetooth is 

a standard for the short-range wireless interconnection of cellular phones, computers, and other 

electronic devices.  

 

Please take some time and read through the pointers included in reference [1] to read more about 

Bluetooth. It is outside the scope of the EECE module at this stage to teach details about it. 

However, you are encouraged to look into it more on your own. 

 

2.2. HC-06 Bluetooth Module 

 

In this lab, we will use the JY-MCU BT_BOARD V1.07 (also referred to as HC-06): Arduino 

Wireless Bluetooth Transceiver Module [2,3], which is shown in Fig.1. 

 

 
Figure 1: Photographs of the "JY-MCU BT_BOARD V1.07" Bluetooth module. 

This is a popular serial Bluetooth module, which uses the Bluetooth Specification v2.0. 

It can work either as a master or a slave. It is built around the BC417 Bluetooth to Serial chip, 

which is a single chip radio and baseband IC for Bluetooth 2.4 GHz systems including enhanced 

data rates (EDR) to 3Mbps [4]. This chip is loaded with features, including UART interface with 

programmable baud rate up to 3Mbaud with an optional bypass mode. By default, it works as a 

Slave, 9600 baudrate, N, 8, 1, and Pincode 1234.  

 

While very popular (there are several other versions, such as HC-03, HC-04, and HC-05 among 

others) and easy to use, it somewhat lacks in up to date documentation. Nevertheless, the manual 

of version 1.05 is a good starting point to read and find out more about this module. Please take 

some time to briefly read the manual available for download at [3]. 

 

Some of the features of the HC-06 Bluetooth module include [3]: 

 Serial port operation voltage 3.3V 

 Baudrate set at 9600, but you can change it using AT commands 

 Operation current, pairing at 30mA, after pairing 8mA during communication 

 Can pair and communicate with laptops, laptop Bluetooth adapters, Bluetooth Shields on 

Arduino, smart phones such as Android, etc. 

 

The module has six pins: 

 VCC: supply voltage with range 3.6 V to 6 V (BC417 chip needs 3.3V generated with the 

help of a voltage regulator on the same PCB). 

 GND: ground. 

 RXD: Serial RX. Note, this would be connected to the TX pin of the Arduino board.  

 TXD: Serial TX. As you would expect, this this would be connected to the RX pin of the 

Arduino board. 

 KEY: Not used. 

 STATE: Not used. 

 

3.  Preparation for Laboratory 3 

 

To be completed PRIOR TO your lab session on 11/14/2017. 

 

Type up the sketch using the code in Appendix A. Take some time and browse the datasheet of 

the HC-06 module: http://dejazzer.com/geen1200/resources/HC06_bluetooth_manual.pdf  

Note what you will use as the identifier for YOUR HC-06 module. 

Note the (default) pincode for the HC-06. 

 

Work on experiment 2 

 

4.  In the Laboratory 

 

Complete the following experiments. As each experiment is completed, show the working 

system to a teaching assistant for credit. 

 

 

http://dejazzer.com/geen1200/resources/HC06_bluetooth_manual.pdf

1) Experiment 1:  Learning about Bluetooth Communication with the HC-06 

 

The first example is completely worked out for you. Its purpose is to get you up to speed with 

using the HC-06 Bluetooth module in a very simple experimental setup.  

 

a)  Hardware setup 

The schematic diagram of this setup is shown in Fig.2.  We will connect the HC-06 to the 

Arduino board, which then sends the message “Cris says hi!” to the HC-06 once per second. This 

transfer is done using a software serial connection between the Arduino board to the HC-06 

module. At the same time, the HC-06 and the host PC (your laptop) are wirelessly connected 

with Bluetooth. The HC-06 module uses this wireless connection to transmit to the laptop the 

message it has received from the Arduino board. On the laptop, we simply display the received 

message (wirelessly via Bluetooth) in a Serial terminal connected to the COM port associated 

with the Bluetooth connection. 
 

 
Figure 2: Schematic diagram of experiment #1. 

 

Connect all the necessary components using the schematic from figure 2.  For the 3 resistor 

converter, you may use three of the 220 Ohm resistors that you used in the previous labs. Your 

setup should look something like that shown in Fig.3. 

Figure 3: Experimental setup for experiment #1. 

 

b)  Software (i.e., sketch) 

Create a new Arduino project which uses the code listed in the sketch from Appendix A. 

Compile and program the Arduino board.  

 Make sure that you replace “cristinel” inside the example sketch with a unique identifier 

for your module in order to be able to identify it in the class-room.  If you don’t give your 

module its own name, everyone’s Bluetooth modules will be displayed with the default name 

of the module, HC-06, and you won’t be able to easily determine which is your own.   

 Consider making the message you will be sending something unique to your 

transmitter/receiver.   
  

c)  Pair your laptop with the Bluetooth module.  Display messages received by module in 

the Serial terminal. 

 Launch the Bluetooth Manager on your laptop. 

Select the Bluetooth icon on bottom-right of screen on a Windows 10 machine – or – select 

the Windows Start icon on the bottom-left of the screen, then select Settings, then Devices, 

then Bluetooth. Select your device and pair. In my case for example, I paired my laptop with 

the “cristinel” device. If you are asked about a Pincode, type: “1234”.  

 Determine the COM port for the HC-06 module 

In the Bluetooth section of the same device manager, select “More Bluetooth Options” (you 

may need to scroll down to see it), then select COM Ports, then note which port is 

“Outgoing” for your device. 

 Launch the Serial terminal. 

In the Arduino IDE, set the COM port to the above, then open the serial terminal. If all 

connections are correct and everything runs ok on the Arduino board, then, you should see in 

the serial terminal what’s transmitted by the HC-06 module: that is, the message “Cris says 

hi!” every second. This message is given to the HC-06 module by the sketch running on the 

Arduino. Fig.4 shows a close up of the serial terminal in my case.   

 

 
Figure 4: Snapshot of serial terminal connected to COM42 associated with the Bluetooth connection to 

the HC-06 module. 

 

2) Experiment 2:  Wireless temperature and humidity data logger 

 

Combine the above experiment with the data acquisition experiment from the previous lab to 

transmit the temperature and humidity data wirelessly to the host PC. The block diagram of this 

experiment is shown in Fig.5. You must write a new Arduino sketch where you can reuse code 

from the previously studied sketches. Wire up the entire system, program the Arduino, and do 

what it takes to make it work! 

Figure 5: Schematic diagram of experiment #2. 

Appendix A – Listing of sketch experiment1.ino  
 

// cristinel ababei; sep.19.2016; copyleft :-) 

// Susan Schneider; sep.29.2016 

// 

// Description: 

// this is a simple Arduino project which uses an HC-06 module 

// to send wirelessly, via Bluetooth, a simple message repeatedly 

// to a host laptop that must have Bluetooth connectivity enabled; 

// in addition, in the beginning only, it also changes the name 

// of the HC-06 module to make it unique to my situation; this 

// helps in a class-room scenario where multiple modules may be on 

// at the same time; 

 

#include <SoftwareSerial.h> 

 

#define DEBUG true 

 

// connect the HC-06 TX to Arduino pin 2 RX; 

// connect the HC-06 RX to Arduino pin 3 TX through a voltage divider created 

// with three identical resistors; 

SoftwareSerial hc06_serial(2, 3); // receiver | transmitter 

 

void setup() 

{ 

  // (1) initialize serial link to HC-06; baudrate for communicating  

  // with the HC-06 is 9600; 

  hc06_serial.begin(9600);  

  // send three AT commands to the HC-06 module; the third one is  

  // the one that changes its name to "cristinel"; here, replace 

  // “cristinel” with your own unique identifier!  

  sendCommand("AT",1000,DEBUG);  

  sendCommand("AT+VERSION",1000,DEBUG);  

  sendCommand("AT+NAMEcristinel",1000,DEBUG); 

 

  // (2) insert 1 sec delay; 

  delay(1000); 

} 

 

void loop()  

{ 

    // (1) send message to HC-06, which will pick it up and transmit via 

    // bluetooth; 

    hc06_serial.println("Cris says hi!");  //PERSONALIZE THIS MESSAGE!!! 

 

    // (2) insert 1 sec delay; 

    delay(1000);  

} 

 

// Name: sendCommand 

// Description: function used to send AT commands to HC-06. 

// Params:  

//    command - the data/command to send;  

//    timeout - the time to wait for a response;  

//    debug - print to Serial window? (true = yes, false = no); 

// Returns: response from the HC-06 bluetooth module (if there is a reponse); 

String sendCommand(String command, const int timeout, boolean debug) 

{ 

    String response = ""; 

            

    hc06_serial.print(command); // send the read character to the hc-06; 

     

    long int time = millis();    

    while ( (time+timeout) > millis()) { 

      while (hc06_serial.available()) {       

        // HC-06 module has data; so display its output to the serial window;  

        char c = hc06_serial.read(); // read the next character 

        response+=c; 

      }   

    } 

     

    if (debug) { 

      // do nothing here to keep sketch simple; 

    }    

    return response; 

} 

 


